

```

enquanto i > 0 faça
  i = i - 0,05;
  imprima (i);
  se i + 1 > 1,45
    então i = i + 0,20;
    senão abandone;
fim se;
fim enquanto;

```

## 4.4 DEFINIÇÃO DE NOVOS TIPOS

Nem sempre os tipos básicos (*inteiro*, *real*, *caracter* e *lógico*) são suficientes para exprimir estruturas de dados em algoritmos. Por exemplo, considere o problema mostrado no Cap. 3, em que um professor com 5 alunos deseja imprimir a nota e média de seus alunos. Foram utilizadas 5 variáveis reais para contar as notas dos cinco alunos. Imagine que o número de alunos da turma seja 80. Só a declaração destas variáveis tornaria impraticável a redação do algoritmo. Daí a necessidade de novos tipos serem criados. Um destes tipos, o *vetor*, será estudado nesta unidade.

Para o problema citado, seria ideal uma estrutura de dados que contivesse todas as notas e pudesse ser referenciado pelo conjunto ou cada nota individualmente:

| NOTAS |     |     |       |     |
|-------|-----|-----|-------|-----|
| 5,0   | 3,0 | 8,0 | ..... | 9,5 |
| 1     | 2   | 3   |       | 80  |

No PORTUGOL, a criação destes tipos pode ser feita segundo a seguinte especificação de tipo:

*tipo v = vetor [li : ls] <tipo básico>;*

onde *li* é limite inferior e *ls* o limite superior. <tipo básico> é um dos tipos básicos já conhecidos. *li* e *ls* só podem ser valores inteiros.

Esta especificação apenas indica um modelo. Para efetivar esta estrutura dentro do algoritmo, é necessário declará-la dando um nome à variável que será criada segundo o modelo especificado.

Por exemplo:

*tipo v = vetor [1 : 80] real;*  
*v: NOTAS;*

O número de elementos de um vetor será dado por  $ls - li + 1$ . Isto significa que as posições do vetor são identificadas a partir de *li*, com incrementos unitários, até *ls*:

|    |        |        |       |    |
|----|--------|--------|-------|----|
| li | li + 1 | li + 2 | ..... | ls |
|----|--------|--------|-------|----|

Exemplo:

Qual é o número de elementos e o que representa a especificação e declaração abaixo?

*tipo vet = vetor [5 : 9] caracter;*  
*vet: NOME;*

Solução:

O vetor tem  $9 - 5 + 1 = 5$  elementos e pode ser visto como:

| NOME |   |   |   |   |
|------|---|---|---|---|
|      |   |   |   |   |
| 5    | 6 | 7 | 8 | 9 |

Cada elemento de um vetor é tratado como se fosse uma variável simples. Para referência a um elemento do vetor utiliza-se o nome do vetor e a identificação do elemento (índice) entre colchetes. Por exemplo, se desejarmos atribuir o valor "FMI" ao elemento identificado pelo índice 6 do vetor anterior, teremos:

NOME [6] ← "FMI" que produzirá:

|   |     |   |   |   |
|---|-----|---|---|---|
|   | FMI |   |   |   |
| 5 | 6   | 7 | 8 | 9 |

A seguir serão dados vários exemplos de manipulação de vetores.

Exemplo 1:

O que será impresso no algoritmo abaixo?

```

início
  inteiro i;
  tipo v = vetor [1 : 6] inteiro;
  tipo c = vetor [1 : 6] caracter;
  v: VE;
  c: CA;
  VE [1] ← 1;
  VE [2] ← 1;
  VE [3] ← 2;
  VE [4] ← 2;
  VE [5] ← 5;
  VE [6] ← 6;
  CA [1] ← "SEG";
  CA [2] ← "TER";
  CA [3] ← "QUA";
  CA [5] ← "SEX";
  CA [6] ← "SAB";
  para i de 1 até 6 passo 2 faça
    imprima (CA [VE [i]]);
  fim para;
  imprima (CA [VE [3]]);
fim

```